

תרגיל 2 - חלק יבש - פתרון

המתרגל האחראי על התרגיל: תומר כץ.

שאלותיכם במייל בעניינים מנהלתיים בלבד, יופנו רק אליו.

כתבו בתיבת **subject**: יבש 2 את"ם.

שאלות בעל-פה ייענו על ידי כל מתרגל.

הוראות הגשה:

- לכל שאלה יש לרשום את התשובה במקום המיועד לכך.
- יש לענות על גבי טופס התרגיל ולהגיש אותו באתר הקורס כקובץ PDF.
- על כל יום איחור או חלק ממנו, שאינו בתיאום עם המתרגל האחראי על התרגיל, יורדו 5 נקודות.
- הגשות באיחור יש לשלוח למייל של אחראי התרגיל בצירוף פרטים מלאים של המגישים (שם+ת.ז).
- שאלות הנוגעות לתרגיל יש לשאול דרך הפיאצה בלבד.
- ההגשה בזוגות.

שאלה 1 (45 נק') – שגרות:


ג'וני סטודנט אחראי כל בוקר בשש 30 כתב קוד אסמבלי. לפניכם מקטע הנתונים שג'וני כתב:

```
1 .section .data
2 A: .long 3
3   .quad B
4   .quad C
5 B: .long 4
6   .quad D
7   .quad 0
8 C: .int 5
9   .quad E
10  .quad F
11 D: .int 7
12   .quad 0
13   .quad 0
```

```
14 E: .int 8
15   .quad 0
16   .quad 0
17 F: .int 9
18   .quad G
19   .quad H
20 G: .int 10
21   .quad 0
22   .quad 0
23 H: .int 11
24   .quad 0
25   .quad 0
26
```

א. ציירו את הגרף המתקבל מפירוש מקטע הנתונים (מומלץ להסתכל בתרגול 3 תרגיל 1 ולהיזכר שם על אופן פירוש הזיכרון כרשימה מקושרת). בכל צומת בגרף ציינו את התווית המתאימה לו בלבד (אין צורך לציין ערכים נוספים) (3 נקודות)

פתרון:



ג'וני לא מפחד משגרה שוחקת ולכן כותב את השגרה func וקוד שמתשמש בה:

```

27 .section .text
28 .global _start
29 _start:
30     mov $8, %esi
31     mov $A, %rdi
32     call func
33     movq $60, %rax
34     movq $0, %rdi
35     syscall
36
37 func:
38     pushq %rbp
39     movq %rsp, %rbp
40     cmp (%rdi), %esi
41     jne continue
42     mov $1, %eax
43     jmp finish
44
45 continue:
46     cmpq $0, 4(%rdi)
47     je next
48     pushq %rdi
49     mov 4(%rdi), %rdi
50     call func
51     pop %rdi
52     cmp $1, %eax
53     je finish
54 next:
55     cmpq $0, 12(%rdi)
56     je fail
57     pushq %rdi
58     mov 12(%rdi), %rdi
59     call func
60     pop %rdi
61     cmp $1, %eax
62     je finish

```

```

63 fail:
64     mov $0, %eax
65 finish:
66     leave
67     ret

```

- ב. נתון שבתחילת התוכנית ערך של `rsp` הוא `x`. כאשר `x` הוא מספר בקסדצימלי. מה הוא הערך המקסימלי ומה הערך המינימלי ש`rsp` יכול לאורך ריצת התוכנית? תנו נוסחא שהמספרים בה הם בבסיס הקסדצימלי (בטאו את התשובה בהאמצעות `x`). (5 נקודות)
- הערך המקסימלי הוא `x` כי המחסנית רק קטנה לאורך התוכנית.
- הערך המינימלי בתוכנית הוא `x-40` נבחין שהוא מתקבל כאשר `rdi` מצביע על `D`. נבחין שכל קריאה לפונקציה דוחפת למעשה שני דברים (בני 8 בתים) למחסנית. הכוונה היא לכותבת החזרה מהשגרה ולערך `rbp` שנדחף בכניסה לפונקציה. בנוסף כל שמתבצעת קריאה ריקורסיבית גם `rdi` נכנס למחסנית ולכן כאשר `A` מבצע קריאה ריקורסיבית של הפונקציה על `B` `rdi` גם נכנס למחסנית ובאופן דומה גם כש `B` עושה קריאה על `D`. מכאן יש עוד שתי דחיפות של 8 בתים למחסנית. נבחין שבעת הקריאה של `D` לא דוחפים את `rdi` כי אין קריאה ריקורסיבית.
- ג. רשמו מה יהיה פלט הפונקציה עבור קטע הקוד הנוכחי (7 נקודות)
- הפלט הוא 1
- ד. המירו את הפונקצייה לשפת `C` על ידי כך שתשלימו את המקומות החסרים בקוד. העיזרו בהגדרת `structn` שנתונה לכם (10 נקודות):
- `structn` הנתון:

```

typedef struct _Node {
    int data;
    struct _Node *left;
    struct _Node *right;
} Node;

```

הערה 1: שני הפרמטרים צריכים להיות תואמים לשני הפרמטרים של פונקציה האסמבלי גם מבחינת תפקיד וגם מבחינת סדר. כלומר, root צריך להתאים בתפקידו לפרמטר הראשון שמועבר לפונקציה בשפת אסמבלי גם מבחינת הקונבנציה שלמדנו.

הערה 2: אורך הקו לא מלמד על אורך האיבר שצריך להשלים. מותר להשלים יותר ממילה אחת בכל קו אך לא יותר מפקודה אחת!

```
__int__ func ( Node *root, __int__ x){
    If (root->data == __x__ )
        return 1;
    if (root->left != null)
        if (func(root->left, x))
            return 1;
    if (root->right != null)
        return func(root->right, x);
    return 0;
}
```

התקבלו גם תשובות לגבי ערך החזרה שהוא bool

הערה: בסעיפים הבאים יש כל מיני שינויים בקוד. כל שינוי מתקיים רק בסעיף בו מופיע. זאת אומרת הסעיפים לא תלויים אחד בשני.

ה. מוני חבר של ג'וני הוא לא כמו ג'וני. הוא אוהב לעשות שינויים רבים בקוד. הוא מחליט לקחת את המקטע הנתונים של ג'וני ולשנות בכל struct את הquad int. כלומר מקטע הנתונים ישתנה כך:

```
1 .data
2 A: .quad 3
3 .quad B
4 .quad C
5 B: .quad 4
6 .quad D
7 .quad 0
```

ובאופן דומה כל שאר האותיות יחליפו את הנתון הראשון במקום בquad int. רשמו את השינויים שצריכים להיות בקוד על מנת שיעבוד בצורה תקינה עם מקטע הנתונים החדש (5 נקודות)

נצטרך לבצע את השינויים הבאים:

כיוון וגודל הנתון בצומת הוא עכשיו 8 בתים כל גישה לבן הראשון תצטרך להיות בהזחה של 8 בתים rdi ולא כ 4 ומכאן כל גישה מהצורה 4(%rdi) תשתנה ל 8(%rdi) (שורות 46 ו 49). ובאופן דומה כל גישה לבן השני של צומת תוחלף מ 12(%rdi) ל 16(%rdi) (שורות 58 ו 61). **בנוסף יש צורך לשנות את השימוש בrsi esi אך לא הורדו נקודות לאלו שלא ציינו שינוי זה**

ו. ג'וני מתחיל להתעייף מהשגרה ומחליט לקום ולשנות את מבנה הנתונים באופן הבא:

```
11 D: .int 7
12 .quad A
```

מה יהיה פלט התוכנית? יש לסמן תשובה מבין התשובות הבאות ולנמק במשפט אחד: (5 נקודות)

- התוכנית תסתיים ופלט הפונקציה יהיה 1
- התוכנית תסתיים ופלט הפונקציה יהיה b
- התוכנית תכנס ללולאה אנסופית
- התוכנית תקרוס במהלך ריצה
- התוכנית כלל לא תבנה

שימו לב לכך שאמנם לוגית יש כאן לולאה אינסופית אבל בכל קריאה ריקורסיבית דוחפים איברים למחסנית.

- ז. פתאום ג'וני כמו מוני! מחליט לבצע שינויים נוספים ולא שגרתיים בקוד מול כל שינוי שג'וני מציע עליכם לכתוב האם נכונות השגרה תיפגע (האם יש קלט עבורו השגרה לאחר השינוי שונה מהשגרה לפני השינוי). הסיברו **בקצרה** את תשובתכם! (10 נקודות)
- a. מחיקת הפקודת `push pop` שבשורות 60 ו 57..
אין שינוי כי אין צורך לשחזר את ערך `rdi` ביציאה מהקריאה הריקורסיבית
- b. מחיקת הפקודה `pop` בשורה 60
אין שינוי כי פקודת `leave` מחזירה את `rsp` למצב של להצביע על כתובת החזרה ובכך בעצם מוציאה מהמחסנית את כל האיברים.
- c. מחיקת `push pop` שבשורות 48 ו 51
עלול להיות שינוי בעיות כי במהלך הקריאות הריקורסיביות ערך `rdi` משתנה ויש צורך לשחזר את ערכו בשביל לסרוק את תת העץ של הבן השני.
- d. הוספת פקודה `push %rdi` אחרי `continue` בשורה 45
אין שינוי מאותה סיבה שב-`b` אין שינוי
- e. הוספת הפקודה `push %rdi` אחרי `continue` בשורה 45, שינוי פוקדת ה-`pop` שבשורה 51 לפקודה: `mov (%rsp), %rdi` ומחיקת הפקודת `push pop` בשורות 60 ו 57.
אין שינוי פקודת ה-`mov` שהוספנו במקום פקודת ה-`pop` משחזרת את `rdi` דבר שמאפשר מעבר על תת העץ השני של הצומת. בנוסף זה שאין `pop` שמוציא את `rdi` לא פוגע בגלל הפקודה `leave`.

שאלה 2 (30 נק') – קריאות מערכת:

ג'ואי מרגיש מתוסכל מכך שחבריו חושבים שהוא פחות חכם מהם. לכן, הוא מחליט להרשים אותם בעזרת כתיבת קוד אסמבלי.

א. לפניכם מקטע הנתונים שג'ואי כתב מבלי ערכי הנתונים עצמם:

```
.section .data
msg1: .ascii ???????
msg2: .ascii ???????
msg1_len: .quad ____
msg2_len: .quad ____
all_msg_len: .quad ____
```

ג'ואי לא יודע עדיין אילו מחרוזות הוא יכתוב. עליכם להשלים את המקומות הריקים שקשורים לאורכי המחרוזות כך שמשתנה `msg1_len` יהיה האורך של `msg1`, `msg2_len` יהיה האורך של `msg2` ובמשתנה `all_msg_len` יהיה שווה לסכום אורכי המחרוזות `msg1` ו-`msg2`. שימו לב עליכם לעשות זאת בצורה כזו שהאורכים יהיו נכונים בעת ריצת התוכנית ללא קשר לאיזה מחרוזות ג'ואי ישים ב-`msg1` וב-`msg2`. (3 נקודות)

פתרון:

```
.section .data
msg1: .ascii ???????
msg2: .ascii ???????
msg1_len: .quad msg2 - msg1
msg2_len: .quad msg1_len - msg2
all_msg_len: .quad msg1_len - msg1
```

ב. כעת נתון מקטע הנתונים שכולל את המחרוזות:

```
.section .data
msg1: .ascii "HOW YOOOU DOOIN?"
msg2: .ascii "JOEY DOESN'T SHARE FOOD!"
msg1_len: .quad ____
msg2_len: .quad ____
all_msg_len: .quad ____
```

לפניכם נתונה התוכנית שג'ואי כתב ומוצגת כאן גם הפונקציה שכתב:

```
Joey_func:
    cmp %rbx, %r9
    je end
    # addb $0x20, (%rsi)
    test $1, %rbx
    jnz skip
    syscall
skip:    inc %rsi
        inc %rbx
        call Joey_func
end:    ret

.section .text
.global _start
_start:
    mov $msg1, %rsi
    mov $1, %rdi
    mov $1, %rdx
    mov $1, %rax
    xor %rbx, %rbx
    movq msg1_len, %r9
    call Joey_func
```

מה יודפס בסיום ריצת הקוד? (שימו לב השורה השלישית בפונקציה נמצאת בהערה ולא רלוונטית לסעיף). (5 נקודות)

HWYUDON

ג. כעת מורידים את הסולמית שנמצאת בפונקציה (וכעת הפקודה חלק מהקוד) בנוסף מחליפים את השורה `movq msg1_len, %r9` בשורה: `movq all_msg_len, %r9`.

הערה: שינויים אלו ילוו אותנו גם בסעיפים הבאים (בסעיפים ד - ו השינויים בסעיף ג עדיין תקפים).

מה יודפס כעת בסיום ריצת הקוד? (5 נקודות)

`hwyoudonje@osG@hr@od`

ד. בזמן שג'ואי אכל בסלון סנדוויץ, חיית המחמד שלו (אפרוח) טיילה על המקלדת והוסיפה את הפקודה:

`inc %r9`. הפקודה נוספה שורה לפני הקריאה לפונקציה של ג'ואי בתוכנית הראשית.

מה יהיה פלט התוכנית כעת? (2 נקודות)

`hwyoudonje@osG@hr@od0`

ה. חברה טובה של ג'ואי פיבי אמרה לו ששימוש ברגיסטר `r9` מביא מזל רע. ג'ואי נלחץ נורא והחליט שיש לבצע שינוי בקוד מבלי לשנות את תוצאות הפעולה של הפונקציה (כלומר הפלט צריך להיות זהה). כיוון ולא ידע איך לשנות את הקוד הוא החליט לבקש את עזרת חבריו.

בסעיף הזה יופיעו העצות של כל החברים. עליכם לרשום ליד כל עצה האם היא לדעתכם תעזור לג'ואי. **נמקו בקצרה (!)** (10 נקודות)

צ'נדלר מציע להחליף את השימוש ב`r9` בשימוש ב`rcx`.

לא יעזור, לאחר הקריאת מערכת `rcx` יכיל את הכתובת של הפקודה הבאה לביצוע אחרי הקריאת מערכת ולכן התוכן שלו ידרס לאחר הקריאה הראשונה. מכאן באיטרציה הבאה ערכו כבר יהיה שונה מערכו המקורי.

מוניקה מציעה להחליף את השימוש ב`r9` בשימוש ב`r11`.

לא יעזור. לאחר הקריאת מערכת ערך `r11` יהיה שווה לערך של אוגר הדגלים לפני קריאת המערכת. לכן, התוכן המקורי של אוגר זה ידרס אחרי הקריאת מערכת הראשונה. מכאן באיטרציה הבאה ערכו כבר יהיה שונה מערכו המקורי.

פיבי מציעה להחליף את השימוש ב`r9` בשימוש ב`rdi`.

לא יעזור, `rdi` קובע לאן תודפס המחרוזת. לכן אסור לדרוס אותו.

רייצ'ל מציעה להחליף את השימוש ב`r9` בשימוש ב`r12`.

השינוי יעזור. אין מניעה להשתמש ברגיסטר זה.

רוס מציע להחליף את השימוש ב`r9` בשימוש ב`rbp`.

יעזור, אין שימוש באוגר הזה ולמעשה הוא לא חייב לבוא בהקשר של מחסנית כלל.

ו. חבריו של ג'ואי מסבירים לו שהשימוש שלו ברקורסיה מיותר ובזבזני והוא יכול את אותו קוד בדיוק לכתוב בלולאות. ג'ואי מחליט לבצע את השינויים הבאים:
בתוכנית הראשית בשורה שלפני ביצוע הפקודה call ג'ואי מוסיף את הפקודה:
`mov $Joey_func, %rcx`

ובתוך הפונקציה ג'ואי מוחק את השורה בה יש שימוש בפקודה call והחליף אותה בפקודה:
`jmp *%rcx`
שימו לב שהתווית end נמצאת אחרי פקודה וז. לצורך הבהרה הפונקציה נראת כך כעת:

```
Joey_func:
    cmp %rbx, %r9
    je end
    addb $0x20, (%rsi)
    test $1, %rbx
    jnz skip
    syscall
skip:    inc %rsi
        inc %rbx
        jmp *%rcx
end:    ret
```

כיצד שינוי זה ישפיע על אופן ריצת הפונקציה. מה יודפס אם נריץ את הפונקציה? (5 נקודות)
האיטרציה הראשונה תתבצע ותודפס האות h אך לאחר הקריאת מערכת הראשונה הערך של rcx ישתנה להיות הכתובת של השורה הבאה אחרי syscall. לכן ביצוע הפקודה: `jmp *%rcx` שקולה למעשה לקפיצה לתווית skip. ומכאן קיבלנו לולאה אינסופית.

שאלה 3 (25 נק') – רמות הרשאה ואוגר הדגלים:

א. הפקודה `pushfq` דוחפת את הערך של אוגר הדגלים למחסנית. והפקודה `popfq` מוציאה את אוגר הדגלים מהמחסנית. הסבירו כיצד באמצעות שילוב של שתי פקודות אלו ניתן להדליק את הדגלים `CF` ו-`OF`. שימו לב במידה ואחד הדגלים כבר דלוק יש להשאירו דלוק כלומר, בסיום התהליך על שני הדגלים להיות דולקים. אין לשנות את שאר הביטים בריגסטר הדגלים. בנוסף, אין לשנות אף רגיסטר שהוא לא `rsp`, `rip`, `rflags` (גם לא באופן זמני). (7 נקודות)
הערה: במידה ובדקתם את עצמכם באמצעות דיבגר וראיתם שנדלק גם דגל `TF` זה בסדר תלמדו בהמשך מדוע הוא נדלק תוך כדי דיבוג.

הכניסו את אוגר הדגלים למחסנית על ידי `pushfq`. כעת נבחין שדגל `OF` הוא ביט מספר 11 ב-`CF` הוא ביט מספר 0. נבצע פקודת `or` בין ראש המחסנית לערך 801 בהקסדצימלי. נבחין שערך זה הוא בבינארי הכל 0 מלבד ביטים מספר 0 ו-11. כתוצאה מהזז ביטים 0 ו-11 בוודאות יהיו דולקים בראש המחסנית ושאר הביטים לא ישתנו. לסיום נבצע `popfq` ובכך נעדכן את אוגר הדגלים.

ב. הולי התחמנית רוצה לאפשר לעצמה גישה ישירה אל התקני הקלט פלט ללא צורך בקריאות מערכת. איזה שינוי באוגר הדגלים יכול לעזור להולי במטרתה? (4 נקודות)
הערה: לא צריך לציין פקודה ספציפית, רק להגיד מה צריך לעשות ברמה התיאורטית

ביטים 12 ו-13 באוגר הדגלים מייצגים את ה-`IOPL`. ה-`IOPL` הוא התנאי על רמת ההרשאה שמשמש צריך כדי לגשת להתקני קלט פלט. אם מתקיים $CPL \leq IOPL$ אז המשתמש יוכל לגשת להתקני הקלט פלט. אם הולי תשנה את ה-`IOPL` להיות הערך המקסימלי (11) כלומר לרמת ההרשאה הנמוכה ביותר, אז התנאי הזה יתקיים גם ברמת ההרשאה הנמוכה ביותר. וכך היא תוכל לגשת ישירות להתקני הקלט פלט.

ג. הולי מחליטה לנסות את התעלול מסעיף א' רק שבמקום לשנות את `CF` ו-`OF` היא רוצה לשנות את ה-`IOPL`. להפתעתה, היא לא מצליחה לשנות את הביטים הללו. הסבירו מה ההגיון בכך שהיא לא מצליחה לשנות את ה-`IOPL`? התייחסו לצורך בקריאות מערכת (4 נקודות)
במידה והיה ניתן לשנות את ה-`IOPL` היה אפשר לגשת ישירות להתקני הקלט פלט מכל רמת הרשאה ולמעשה לא היה צורך בקריאות מערכת. גישה ישירה להתקני קלט פלט הייתה פותחת שער למשתמשים זדוניים לפגוע בכלל התוכניות במחשב ולא רק בתוכנית שהם עצמם מריצים.

ד. וולי החבר המבולבל של הולי מתלבט כיצד ניתן לחסום פסיקות תוכנה לכן הוא שואל את הולי. אילו מבין התשובות הבאות על הולי לענות לו? יש לסמן את האפשרות הנכונה. (5 נקודות)

1. כיבוי דגל `IF` באוגר הדגלים
2. הדלקת דגל `IF` באוגר הדגלים
3. שינוי `CPL` ל-00
4. לא ניתן לחסום פסיקות תוכנה.

לא ניתן לחסום פסיקות תוכנה כיוון והן מגיעות מבפנים בצורה יזומה של המעבד. כלומר המעבד הוא זה שמבקש את הפסיקה ולא ממשיך בלעדיה ומכאן אין טעם לחסום את הפסיקה. תשובה נכונה היא 4.

ה. כעת נתון שוולי הצליח להגיע למצב שבו CPL שווה ל0. וולי מעוניין לחסום פסיקות חומרה שאינן מועברות דרך כניסת NMI. כיצד הוא יכול לעשות זאת? (5 נקודות)

1. כיבוי דגל IF באוגר הדגלים

2. הדלקת דגל IF באוגר הדגלים

3. עליו לחבר את הפסיקות לכניסת NMI ואז לכבות את דגל IF

4. לא ניתן לחסום פסיקות חומרה ולכן לא יצליח.

הפסיקות שאינן מועברות דרך כניסת NMI כן ניתנות לחסימה. אלו שמחוברות לכניסת NMI הן אלו שלא ניתנות לחסימה (הפסיקות שמדווחות על בעיות חומרה קריטיות).הדרך לחסום היא באמצעות כיבוי דגל IF.